```
> # Quick Maple Primer for Physics Students
  # By Miles V. Aronnax
  # Updated: 10 June 2008
```

```
> # Indefinite Integrals
  # Say you want to integrate e to the k*x with respect to x...
  int( exp(k*x), x );
```

$$\frac{\mathbf{e}^{(k\,x)}}{k}$$

```
> # Definite Integrals
  # What if I wanted to integrate e to the k*x from -2 to 73.1?
  int( exp(k*x), x=-2..73.1 );
```

$$-\frac{1.\,(\mathbf{e}^{(-2.\,k)} - 1.\,\mathbf{e}^{(73.10000000\;\;k)})}{k}$$

```
> # Numeric Answers
  # I want my answer to be a decimal number, not a hideous ratio
  # or other complicated form. Oh, and k=0.1...
  k := 0.1;
  evalf( int( exp(k*x), x=-2..73.1 ) );
```

$$k := 0.1$$

$$14943.58458$$

```
> # Defining Functions
  # I'm tired of typing that function, can't I just define it?
  SomeStupidFunction := exp(k*x);
  evalf( int( SomeStupidFunction, x=-2..73.1 ) );
  # Note: k has already been defined above, hense the
  #       function is now exp(0.1 * x).
```

$$SomeStupidFunction := \mathbf{e}^{(0.1\,x)}$$

$$14943.58458$$

```
> # Derivatives
  # Okay, now take the derivative of a function with respect to x.
  # Oh, and let's use a new function...
  SomewhatEvilFunction := exp( a*x^2 - b*x*y + c*y^2 + 42 );
  diff( SomewhatEvilFunction, x );
```

$$SomewhatEvilFunction := \mathbf{e}^{(a\,x^2 - b\,x\,y + c\,y^2 + 42)}$$

$$(2\,a\,x - b\,y)\,\mathbf{e}^{(a\,x^2 - b\,x\,y + c\,y^2 + 42)}$$

```
> # Evaluating Derivatives
  # Note: Use the eval function to test a value that is
  #       variable. If we set x:=2 and y:=3.14, then the
  #       function would be a constant value and
  #       we can't take the derivative of a constant.
  #
  #       Also, if this function only had one variable,
  #       then {x=2, y...} could be replaced with x=2
  #       (ie, the curly braces are needed if defining
  #       multiple variables in a single eval command).
  eval( diff( SomewhatEvilFunction, x ), {x=2, y=Pi, a=1, b=2, c=3} );
```

$$(4 - 2\,\pi)\,\mathbf{e}^{(46 - 4\,\pi + 3\,\pi^2)}$$

```
> # Multiple Derivatives
  # How do I take multiple derivatives, either of the
  # same variable or of different ones?
  # Double derivative in x and one derivative in y:
```

```
diff( SomewhatEvilFunction, x, x, y );
```

$$2\,a\,(-b\,x+2\,c\,y)\,\mathbf{e}^{(a\,x^2-b\,x\,y+c\,y^2+42)}-2\,(2\,a\,x-b\,y)\,\mathbf{e}^{(a\,x^2-b\,x\,y+c\,y^2+42)}\,b+(2\,a\,x-b\,y)^2\,(-b\,x+2\,c\,y)\,\mathbf{e}^{(a\,x^2-b\,x\,y+c\,y^2+42)}$$

```
> # Differential Equations
  # I need to solve a differential equation, how do I do it?
  MyDiffEquation := m*diff( x(t), t, t ) + b*diff( x(t), t) + K*x(t) = 0;
  dsolve( MyDiffEquation );
  # Yes, the solution has undefined constants _C1 and _C2 which need
  # to be determined using boundary conditions.
```

$$MyDiffEquation := m\left(\frac{d^2}{dt^2}x(t)\right)+b\left(\frac{d}{dt}x(t)\right)+K\,x(t)=0$$

$$x(t)=\_C1\,\mathbf{e}^{\left(-\frac{\left(b-\sqrt{b^2-4\,K\,m}\right)t}{2\,m}\right)}+\_C2\,\mathbf{e}^{\left(-\frac{\left(b+\sqrt{b^2-4\,K\,m}\right)t}{2\,m}\right)}$$

```
> # Systems of Differential Equations
  # I've got a coupled system, and I just want it solved.
  AnnoyingSystem := [ diff( x(t), t ) = y(t), diff( y(t), t ) = -x(t) ];
  dsolve( AnnoyingSystem );
```

$$AnnoyingSystem := \left[\frac{d}{dt}x(t)=y(t),\ \frac{d}{dt}y(t)=-x(t)\right]$$

$$\{y(t)=\_C1\cos(t)-\_C2\sin(t),\ x(t)=\_C1\sin(t)+\_C2\cos(t)\}$$

```
> # Transcendental Equations / Numeric Roots
  # Got a hideous equation that you can't get an exact root out of?
  HideousEQ := x = cos(x);
  fsolve( HideousEQ );
```

$$HideousEQ := x=\cos(x)$$

$$0.7390851332$$

```
> # 2-D Graphing
  # Note: Right-clicking on plots will give you various options.
  plot( sin(x)/x, x);
```




```
> # 3-D Graphing
  # Note: Like 2-D graphs, right-click on the plot for options.
  #       Additionally, you can rotate 3-D graphs by left-clicking
```

```
#       on the plot and draging the image arround for a better view.
plot3d( sin(x)*y - cos(y)*x, x=-Pi..Pi, y=-Pi..Pi, axes=FRAME );
```




```
> # Series Summations
  # Note: Try summing this equation from 0 to infinity.
  sum( x^n/n!, n=0..4 );
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4$$

```
> # Evaluating Series Summations
  # Note: If you set x=1, not 1.0, then you get an
  #       exact ratio--your choice: decimal number
  #       or ratio. If all else fails, you can force
  #       a ratio to be translated into a decimal by
  #       wrapping the whole command in evalf().
  eval( sum( x^n/n!, n=0..10 ), x=1.0 );
```

2.718281803

```
> # Multiple Plots, One Graph
  plot( [sin(x), sin(x + Pi/7), sin(x + Pi/4)], x );
```




```
> # Infinite Square Well
  # Below are plots of the antisymmetric and symmetric wave functions
  # for two non-interacting particles in a square well. One particle
  # is in the ground state, the other is in the first excited state.
  # Symmetric vs antisymmetric can be judged by the symmetry along y=x.
  AntiSymmetricWaveFunction := cos(Pi*x)*sin(2*Pi*y) -
  sin(2*Pi*x)*cos(Pi*y);
  plot3d( AntiSymmetricWaveFunction, x=-0.5..0.5, y=-0.5..0.5, axes=frame )
  SymmetricWaveFunction := cos(Pi*x)*sin(2*Pi*y) + sin(2*Pi*x)*cos(Pi*y);
  plot3d( SymmetricWaveFunction, x=-0.5..0.5, y=-0.5..0.5, axes=frame );
```

$$AntiSymmetricWaveFunction := \cos(\pi x)\sin(2\pi y) - \sin(2\pi x)\cos(\pi y)$$



$$SymmetricWaveFunction := \cos(\pi x)\sin(2\pi y) + \sin(2\pi x)\cos(\pi y)$$




```
> # Solve n linear equations in n variables
  # Example uses:
  #  x + y + z = 10
  #      y + 2z = 12
  #  2x +    z = 9
  with(linalg);
  gaussjord(matrix([[1,1,1,10],[0,1,2,12],[2,0,1,9]]));
```

[*BlockDiagonal, GramSchmidt, JordanBlock, LUdecomp, QRdecomp, Wronskian, addcol, addrow, adj, adjoint, angle, augment, backsub, band, basis, bezout, blockmatrix, charmat, charpoly, cholesky, col, coldim, colspace, colspan, companion, concat, cond, copyinto, crossprod, curl, definite, delcols, delrows, det, diag, diverge, dotprod, eigenvals, eigenvalues, eigenvectors, eigenvects, entermatrix, equal, exponential, extend, ffgausselim, fibonacci, forwardsub, frobenius, gausselim, gaussjord, geneqns, genmatrix, grad, hadamard, hermite, hessian, hilbert, htranspose, ihermite, indexfunc, innerprod, intbasis, inverse, ismith, issimilar, iszero, jacobian, jordan, kernel, laplacian, leastsqrs, linsolve, matadd, matrix, minor, minpoly, mulcol, mulrow, multiply, norm, normalize, nullspace, orthog, permanent, pivot, potential, randmatrix, randvector, rank, ratform, row, rowdim, rowspace, rowspan, rref, scalarmul, singularvals, smith, stackmatrix, submatrix, subvector, sumbasis, swapcol, swaprow, sylvester, toeplitz, trace, transpose, vandermonde, vecpotent, vectdim, vector, wronskian*]

$$\begin{bmatrix} 1 & 0 & 0 & \frac{7}{3} \\ 0 & 1 & 0 & \frac{10}{3} \\ 0 & 0 & 1 & \frac{13}{3} \end{bmatrix}$$

```
>
```